

Practica 12. Manejo de Archivos de tipo (registros).

Objetivos

Al finalizar esta práctica el estudiante será capaz de:

1. Usar registros para leer y manipular información.
2. Usar archivos de tipo para almacenar y recuperar información por registros.
3. Crear una aplicación que permita guardar y recuperar los datos de un libro (Cota, Autor, Título y Editorial).

Antes de comenzar

Realicemos el Análisis E-P-S de la Aplicación Encuesta.

Entrada.

Dato	Identificador	Tipo	Restricciones
Cota del Libro	Cota	Alfanumérico	<> ""
Autor(es) del Libro	Autor	Alfanumérico	<> ""
Título del Libro	Título	Alfanumérico	<> ""
Editorial del Libro	Editorial	Alfanumérico	<> ""

Tabla 12-1. Análisis EPS de la aplicación "Biblioteca".

Proceso

- Para guardar un registro en el archivo se introducen los datos y se realiza una búsqueda secuencial del campo clave del registro (COTA) sobre todo el archivo. Si éste se encuentra (existe) se muestra un mensaje de error en caso contrario se anexa el registro al final del archivo.
- Para recuperar un registro del archivo se introduce la cota (clave del registro), la cual se busca secuencialmente en todo el archivo y si se encuentra se muestra el registro en los cuadro de texto y en caso contrario se muestra un mensaje de error.

- Para la búsqueda secuencial se abre el archivo y luego se va leyendo uno a uno los registros hasta conseguir el deseado o hasta que llegue al final del archivo.

Salida

El registro de datos.

Para la interfaz de usuario se usaran los siguientes controles:

- 6 etiquetas (Label)
- 4 cajas de texto (TextBox)
- 3 botones de comando (CommandButton)

Actividades a realizar:

1. Ejecute el Visual Basic 6.0, seleccione “*Exe estándar*” en el cuadro de dialogo “*Nuevo Proyecto*”.
2. Establezca las siguientes propiedades al formulario:

Propiedad	Valor
Caption	Practica 12. Archivos de tipo
Forecolor	A su gusto
Height	5600
StartPosition	2 - CenterScreen
Width	8100

Tabla 12-2. Propiedades del formulario de la aplicación “Biblioteca”.

3. Coloque sobre el formulario las seis etiquetas, los cuatro cuadro de texto y los tres botones de comando y fije las propiedades contenidas en la tabla 12-3, de tal manera que se vea aproximadamente como se muestra en la figura 12-1:

Figura 12-1. Formulario 1 de la aplicación “Biblioteca”.

Control	Propiedad	Valor
Etiqueta 1	(Nombre)	Label1
	Caption	Sistema de Información de la Biblioteca Nacional
	Width, Height, Top y Left	7300, 375, 120, 250
	Alignment	2 – Center
	Font	MS Sans Serif , Negrita cursiva, 14
Etiqueta 2	(Nombre)	Label2
	Caption	Sub-Sistema de Información de Libros
	Alignment	2 – Center
	Width, Height, Top y Left	7300, 375, 600, 250
	Font	MS Sans Serif , Negrita cursiva, 12
Etiqueta 3	(Nombre)	Label3
	Caption	Cota:
	Width, Height, Top y Left	1100, 260, 1450, 600
	Alignment	1 – Right Justify
	Font	MS Sans Serif , Negrita, 12

Etiqueta 4	(Nombre)	Label4
	Caption	Título:
	Width, Height, Top y Left	1100, 260, 2050, 600
	Alignment	1 – Right Justify
	Font	MS Sans Serif , Negrita, 12
Etiqueta 5	(Nombre)	Label5
	Caption	Autor:
	Width, Height, Top y Left	1100, 260, 2650, 600
	Alignment	1 – Right Justify
	Font	MS Sans Serif , Negrita, 12
Etiqueta 6	(Nombre)	Label5
	Caption	Editorial:
	Width, Height, Top y Left	1000, 260, 3250, 600
	Alignment	1 – Right Justify
	Font	MS Sans Serif , Negrita, 12
Cuadro de texto 1	(Nombre)	Text1
	Text	
	Width, Height, Top y Left	2000, 375, 1400, 1700
	Font	MS Sans Serif, Normal, 12
Cuadro de texto 2	(Nombre)	Text2
	Text	
	Width, Height, Top y Left	5300, 375, 2000, 1700
	Font	MS Sans Serif, Normal, 12
Cuadro de texto 3	(Nombre)	Text3
	Text	
	Width, Height, Top y Left	5300, 375, 2600, 1700
	Font	MS Sans Serif, Normal, 12
Cuadro de texto 4	(Nombre)	Text3
	Text	
	Width, Height, Top y Left	5300, 375, 3200, 1700
	Font	MS Sans Serif, Normal, 12
Botón 1	(Nombre)	Command1
	Caption	&ANEXAR
	Width, Height, Top y Left	1800, 500, 4450, 1600
Botón 2	(Nombre)	Command2
	Caption	&CONSULTAR
	Width, Height, Top y Left	1800, 500, 4450, 3800

Botón 3	(Nombre)	Command3
	Caption	&SALIR
	Width, Height, Top y Left	1800, 500, 4450, 6000

Tabla 12-3. Propiedades de los controles del formulario de la aplicación "Biblioteca".

4. Inserte el siguiente código de programa en el formulario, en primer lugar el botón Anexar y luego Consultar.

Option Explicit

```

Private Sub Command1_Click()
    Dim RLibro As Tipo_Libro
    Dim CotaB As String * 15
    Dim Enc As Boolean

    ' Captura y VALIDACION de los Campos
    CotaB = Text1.Text

    ' Buscar el libro en el archivo

    Enc = False
    Open "PRUEBA.DAT" For Random As #1 Len = Len(RLibro)
    While (Not Enc) And (Not EOF(1))
        Get #1, , RLibro
        If RLibro.Cota = CotaB Then
            Enc = True
        End If
    Wend

    If Enc Then
        MsgBox "Cota ya Existe. Verifique", vbOKOnly, "Error"
        Text1.SetFocus
    Else
        ' Llenar el Registro
        RLibro.Cota = CotaB
        RLibro.Titulo = Text2.Text
        RLibro.Autor = Text3.Text
        RLibro.Editorial = Text4.Text
        Put #1, , RLibro
    End If
    Close #1
End Sub

Private Sub Command2_Click()
    Dim RLibro As Tipo_Libro

```

```

Dim CotaB As String * 15
Dim Enc As Boolean

' Captura y VALIDACION de los Campos
CotaB = Text1.Text

' Buscar el libro en el archivo

Enc = False
Open "PRUEBA.DAT" For Random As #1 Len = Len(RLibro)
While (Not Enc) And (Not EOF(1))
    Get #1, , RLibro
    If RLibro.Cota = CotaB Then
        Enc = True
    End If
Wend

If Not Enc Then
    MsgBox "Cota no Existe. Verifique", vbOKOnly, "Error"
    Text1.SetFocus
Else
    ' Llenar el Registro
    Text2.Text = RLibro.Titulo
    Text3.Text = RLibro.Autor
    Text4.Text = RLibro.Editorial
End If
Close #1

End Sub

Private Sub Command3_Click()
    End
End Sub

```

5. Inserte un módulo al proyecto y en el mismo coloque la siguiente declaración del Tipo Registro Libro.

```

' Declaración del registro de Datos
Type Tipo_Libro
    Cota As String * 15
    Titulo As String * 30
    Autor As String * 20
    Editorial As String * 20
End Type

```

6. Almacene la aplicación tal y como se ha hecho en las prácticas anteriores.

7. Ejecute la aplicación y verifique su funcionamiento
8. Como complemento de la práctica agregue un botón realizar un listado de todos los libros contenidos en el archivo.